

# NAG Library Routine Document

## G05TCF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

G05TCF generates a vector of pseudorandom integers from the discrete geometric distribution with probability  $p$  of success at a trial.

### 2 Specification

```
SUBROUTINE G05TCF(MODE, N, P, R, LR, STATE, X, IFAIL)
INTEGER          MODE, N, LR, STATE(lstate), X(N), IFAIL
double precision P, R(LR)
```

### 3 Description

G05TCF generates  $n$  integers  $x_i$  from a discrete geometric distribution, where the probability of  $x_i = I$  (a first success after  $I$  trials) is

$$P(x_i = I) = p \times (1 - p)^{(I-1)}, \quad I = 1, 2, \dots$$

The variates can be generated with or without using a search table and index. If a search table is used then it is stored with the index in a reference vector and subsequent calls to G05TCF with the same parameter value can then use this reference vector to generate further variates. If the search table is not used (as recommended for small values of  $p$ ) then a direct transformation of uniform variates is used.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05TCF.

### 4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

### 5 Parameters

- 1: MODE – INTEGER *Input*
- On entry:* a code for selecting the operation to be performed by the routine.
- MODE = 0  
Set up reference vector only.
- MODE = 1  
Generate variates using reference vector set up in a prior call to G05TCF.
- MODE = 2  
Set up reference vector and generate variates.
- MODE = 3  
Generate variates without using the reference vector.
- Constraint:* MODE = 0, 1, 2 or 3.

- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the number of pseudorandom numbers to be generated.  
*Constraint:*  $N \geq 0$ .
- 3: P – *double precision* *Input*  
*On entry:* the parameter  $p$  of the geometric distribution representing the probability of success at a single trial.  
*Constraint:* *machine precision*  $\leq P \leq 1.0$  (see X02AJF).
- 4: R(LR) – *double precision* array *Communication Array*  
*On entry:* if MODE = 1, the reference vector from the previous call to G05TCF.  
 If MODE = 3, R is not referenced by G05TCF.  
*On exit:* the reference vector.
- 5: LR – INTEGER *Input*  
*On entry:* the dimension of the array R as declared in the (sub)program from which G05TCF is called.  
*Suggested value:*  
     if MODE  $\neq$  3, LR = 8 + 42/P approximately (see Section 8);  
     otherwise LR = 1.  
*Constraints:*  
     if MODE = 0 or 2, LR  $\geq$  30/P + 8;  
     if MODE = 1, LR should remain unchanged from the previous call to G05TCF.
- 6: STATE(*lstate*) – INTEGER array *Communication Array*  
**Note:** *lstate* = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.  
*On entry:* contains information on the selected base generator and its current state.  
*On exit:* contains updated information on the state of the generator.
- 7: X(N) – INTEGER array *Output*  
*On exit:* the  $n$  pseudorandom numbers from the specified geometric distribution.
- 8: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry  $IFAIL = 0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by  $X04AAF$ ).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry,  $MODE \neq 0, 1, 2$  or  $3$ .

$IFAIL = 2$

On entry,  $N < 0$ .

$IFAIL = 3$

On entry,  $P < 0.0$  or  $P > 1.0$ ,  
or  $MODE = 0$  or  $2$  and  $P$  is so small that  $LR$  would have to be larger than the largest representable integer. Use  $MODE = 3$  in this case.

$IFAIL = 4$

On entry,  $P$  is not the same as when  $R$  was set up in a previous call to  $G05TCF$  with  $MODE = 0$  or  $2$ .

On entry, the  $R$  vector was not initialized correctly or has been corrupted.

$IFAIL = 5$

On entry,  $LR$  is too small when  $MODE = 0$  or  $2$ .

$IFAIL = 6$

On entry,  $STATE$  vector was not initialized or has been corrupted.

## 7 Accuracy

Not applicable.

## 8 Further Comments

The time taken to set up the reference vector, if used, increases with the length of array  $R$ . However, if the reference vector is used, the time taken to generate numbers decreases as the space allotted to the index part of  $R$  increases. Nevertheless, there is a point, depending on the distribution, where this improvement becomes very small and the suggested value for the length of array  $R$  is designed to approximate this point.

If  $P$  is very small then the storage requirements for the reference vector and the time taken to set up the reference vector becomes prohibitive. In this case it is recommended that the reference vector is not used. This is achieved by selecting  $MODE = 3$ .

## 9 Example

This example prints 10 pseudorandom integers from a geometric distribution with parameter  $p = 0.001$ , generated by a single call to  $G05TCF$ , after initialization by  $G05KFF$ .

## 9.1 Program Text

```

*      G05TCF Example Program Text
*      Mark 22 Release. NAG Copyright 2006.
*      .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MSEED, MSTATE, N, LR
      PARAMETER        (MSEED=1,MSTATE=633,N=10,LR=40000)
*      .. Local Scalars ..
      DOUBLE PRECISION P
      INTEGER          GENID, I, IFAIL, LSEED, LSTATE, SUBID
*      .. Local Arrays ..
      DOUBLE PRECISION R(LR)
      INTEGER          SEED(MSEED), STATE(MSTATE), X(N)
*      .. External Subroutines ..
      EXTERNAL        G05KFF, G05TCF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G05TCF Example Program Results'
      WRITE (NOUT,*)
*      Set the distribution parameter P
      P = 0.001D0
*      Initialize the seed
      SEED(1) = 1762543
*      GENID and SUBID identify the base generator
      GENID = 1
      SUBID = 1
*      Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 20
      END IF
*      Generate the variates
*      Choose MODE = 3 because P is close to 0
      IFAIL = 1
      CALL G05TCF(3,N,P,R,LR,STATE,X,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 20
      END IF
*      Display the variates
      WRITE (NOUT,99999) (X(I),I=1,N)
*
      20 CONTINUE
*
99999  FORMAT (1X,I12)
99998  FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997  FORMAT (1X,' ** G05TCF returned with IFAIL = ',I5)
      END

```

## 9.2 Program Data

None.

## 9.3 Program Results

G05TCF Example Program Results

```

      451
      2238
      292
      225
      2256

```

708  
955  
239  
696  
397

---